

Lab::Measurement – measurement control with Perl

M. Schambeck, E. Fabrizzi, F. Weinelt, S. Reinhardt, and A. K. Hüttel

Institute for Experimental and Applied Physics, University of Regensburg, 93040 Regensburg, Germany

Flexible measurement needed?!

- Tired of following your wires in square meters of LabVIEW diagrams?
- Tired of clumsy string handling and low-level driver functions in your looong C program?
- Use a text processing language to manage your measurement! Use Perl!

```
#!/usr/bin/env perl
# Read out SR830 Lock In Amplifier at GPIB address 13
use 5.010;
use Lab::Moose;

my $lia = instrument(
    type          => 'SR830',
    connection_type  => 'LinuxGPIB',
    connection_options => { pad => 13 }
);

my $amp = $lia->get_amplitude();
say "Reference output amplitude: $amp V";

my $freq = $lia->get_freq();
say "Reference frequency: $freq Hz";

my $r_phi = $lia->get_rphi();
my ($r, $phi) = ($r_phi->(r), $r_phi->(phi));
say "Signal: r=$r V, phi=$phi degree";
```

Currently supported hardware



Hardware driver backends:

- NI-VISA (MS Windows) and all hardware supported by it
- LinuxGPIB and all hardware supported by it
- TCP connection, generic network socket
- USB-TMC lightweight driver (Linux, libusb)
- VXI-11 lightweight driver (Linux, libtirpc)
- Oxford Instruments IsoBus
- Zurich Instruments LabOne API

Growing number of high-level drivers (more are very easy to add):

- Multimeters: HP / Agilent / Keysight
- DC sources: Yokogawa / Keithley / Keysight
- AW generators: Rigol
- Lock-in amplifiers: Stanford Research / Signal Recovery / Zurich Instruments / SyncTek
- Temperature controllers: Lakeshore / Oxford Instruments
- RF / microwave sources, spectrum analyzers, VNAs: Rohde & Schwarz / HP / Agilent / Rigol
- Nanonis Tramea
- and many more...

Key facts

- Open source / free software
- <https://www.labmeasurement.de/>
- License: same as Perl (GPL-1+ or Artistic)
- Releases on CPAN, development on Github
- Contributors and cooperations welcome!



Real world measurement

- Ferromagnetic resonance measurement with vector network analyzer (VNA)
- Outer loop: continuous sweep of magnetic field with OI Mercury iPS magnet controller
- Inner loop: VNA transmission measurement at multiple discrete microwave frequencies

```
#!/usr/bin/env perl
use 5.010;
use Lab::Moose;

# Define the two instruments
my $sips = instrument(
    type          => 'OI_Mercury::Magnet',
    connection_type  => 'Socket',
    connection_options => { host => '192.168.3.15' },
);

my $vna = instrument(
    type          => 'RS_ZVA',
    connection_type  => 'VXI11',
    connection_options => { host => '192.168.3.27' },
);

# Set VNA's IF filter bandwidth (Hz)
$vna->sense_bandwidth_resolution( value => 1 );

# The outer sweep: continuous magnetic field sweep
my $field_sweep = sweep(
    type          => 'Continuous::Magnet',
    instrument => $sips,
    from      => 2,   # Tesla
    to       => 0,   # Tesla
    rate     => 0.01, # Tesla/min
    start_rate => 1, # Tesla/min (rate to approach start point)
    interval => 0,   # run slave sweep as often as possible
);

# The inner sweep: set frequency to 1GHz, 2GHz, ..., 10GHz
my $freq_sweep = sweep(
    type          => 'Step::Frequency',
    instrument => $vna,
    from      => 1e9,
    to       => 10e9,
    step     => 1e9
);

# The data file: gnuplot-style, VNA data prefixed with B
my $datafile = sweep->datafile(
    columns => [ 'B', 'f', 'Re', 'Im', 'r', 'phi' ] );
# Add a live plot of the transmission amplitude
$datafile->add_plot( x => 'B', y => 'r' );
# Define the measurement instructions per (B,f) point
my $meas = sub {
    my $sweep = shift;
    say "frequency f: ", $sweep->get_value();
    my $field = $sips->get_field();

    # this is not really a VNA "sweep", but only a
    # point measurement at one frequency
    my $pd1 = $vna->sparam_sweep( timeout => 10 );
    # Record the result, prefixed with B
    $sweep->log_block(
        prefix => { field => $field },
        block => $pd1
    );
};

# And go!
$field_sweep->start(
    slave      => $freq_sweep,
    datafile   => $datafile,
    measurement => $meas,
    folder     => 'Magnetic_Resonance',
);
```

Output files


```
simon@dilfridge ~/FMR$ ls Magnetic_Resonance_001/
META.yml data.dat data.png magnet-resonance.pl
```

- **META.yml**: various metadata (host, user, date, L::M version, command line arguments)
- **data.dat**: measured data, in tab-separated Gnuplot format
- **data.png**: live plot at the end of the measurement, as a png image
- **magnetic-resonance.pl**: archival copy of the measurement script

Advanced sweep features

- Multidimensional sweeps, e.g. 3D sweep: creating one 2D datafile for each step of the outermost sweep
- Log arrays and matrices of data (PDLs). Useful for spectrum analyzers, VNAs, oscilloscopes which do fast, internally controlled sweeps.
- Extensive support for live plots via gnuplot: line plots (2D data) and color maps (3D data)
- Customizing live plots: access to all gnuplot plot and curve options via PDL::Graphics::Gnuplot


Layer structure



- Modular structure. Easy to extend with new instrument drivers and connection types
- Abstract IO layer, makes instrument drivers independent of hardware backends

High-level sweep framework

- Modern Perl implementation; use state of the art object-oriented programming



- Separate classes for sweeps, datafiles, datafolders, and plots
- Most operational details of sweeps implemented in subclasses of Lab::Moose::Sweep
- High modularity: very easy to extend

New improvements / additions

- VISA::USB connection: USB-TMC over NI-VISA
- Enable magnet persistent mode usage
- RFC3161 timestamps for measurement data
- New drivers
 - Andeen-Hagerling AH2700A ultra-precision capacitance bridge
 - HP 8596E, Rigol DSA815 spectrum analyzer
 - SignalRecovery 7265, Zurich Instruments HF2LI lock-in
 - Oxford Instruments Triton and ITC503 temperature control
 - Keithley 2400, 2450 sourcemeter
 - HP 34420A Nanovoltmeter
 - Tektronix TBS1000C, Keysight DSOS604A oscilloscope
 - Rigol DG5000 series waveform generator
 - Rohde & Schwarz ZNL network analyzer
 - Nanonis Tramea (work in progress)

Reference / Cite as

“Lab::Measurement — a portable and extensible framework for controlling lab equipment and conducting measurements”, S. Reinhardt et al., Comp. Phys. Comm. 234, 216 (2019)